
Table 5: When both SEM_BK and PS_AIR assumptions hold. 
 
5.a. Mplus input for data generation 
   
title: Data generation (true) model for Table 4 
       All the assumptions of the SEM_BK and PS_AIR models hold. That is, 
       exclusion restriction, monotonicity, ignorability, constant effect hold. 
       See Table 2 for PS parameters considered in data generation. 
 
montecarlo: 
      names = y z; ! y is a continuous outcome. 
      nobservations = 500; ! N=500 in each data set. 
      nreps = 500; !500 data sets generated. 
      cutpoints = z(0); ! z is the intervention assignment. 
                        ! split at zero in the normal distribution. 
                        ! 50% control, 50% intervention. 
      genclasses = c(3); ! Under monotonicity, 3 principal strata are generated. 
      classes = c(1); ! analysis is done assuming one class. 
      repsave = all; ! all 500 data sets will be saved. 
      save = all*.dat; ! each data set has a name starting with all. 
 
model population: ! data generation model. 
      %overall% 
       [y*1];  ! generate the outcome. 
        y*1;   ! normally distributed with mean=1 and variance=1. 
       [z*1];  ! generate an intervention assignment variable. 
        z*0;   ! normally distributed with mean=0 and variance=1. 
               ! this variable is dichotomized (see cutpoints command).  
      y on z*0.2; ! the overall treatment assignment effect = 0.2. 
 
     [c#1*0.693147]; ! multinomial logit intercepts  
     [c#2*0.693147]; ! to match true principal stratum proportions (0.4,0.4,0.2). 
  
      %c#1% ! never-improver 
          [y*1] ; ! mean outcome for never-improvers under the control condition. 
           y on z*.0; ! treatment assignment effect for never-improvers = 0. 
      %c#2% ! forward-improver 
          [y*1] ; ! mean outcome for forward-improvers under the control condition. 
           y on z*.5; ! treatment assignment effect for forward-improvers = 0.5. 
      %c#3% ! always-improver 
          [y*1.5] ; ! mean outcome for always-improvers under the control condition. 
           y on z*.0; ! treatment assignment effect for always-improvers = 0. 
 
 Analysis: ! data generation model. 
      type = mixture ; ! data are analyzed assuming a mixture distribution. 
                       ! for Table 4, analyses done here are not necessary. 
                       ! necessary only to complete data generation in Mplus.  
 model: 
        %overall% 
       y on z*0.2; ! total effect of treatment assignment estimated. 
                  ! existence of 3 principal strata is ignored (see classes=1 command). 
 
 
 



Table 5 
5.b. Mplus input for SEM_BK model 
 
title: Monte Carlo simulation using externally generated data.   
data:   file is alllist.dat; type=montecarlo; ! read in 500 data sets from 4.a 
variable: names are y z c3; ! c3 is principal stratum class, z is treatment assignment. 
          usev  are y z m;  ! observed mediator created based on c3 and z (see below). 
define: if (c3==1 and z==1)  then m=0; 
        if (c3==1 and z==0)  then m=0; 
        if (c3==2 and z==1)  then m=1; 
        if (c3==2 and z==0)  then m=0; 
        if (c3==3 and z==1)  then m=1; 
        if (c3==3 and z==0)  then m=1; 
 
model: ! SEM_BK analysis model 
      y on z*0 (c); ! SEM parameter c. true values specified on the right side of *         
                    ! so that coverage rates can be calculated. 
      y on m*.5 (b); ! SEM parameter b. 
      m on z*.4 (a); ! SEM parameter a. 
     [m*0.2] (alpha_m); ! SEM parameter alpha_m. 
 
model constraint: ! this command lets us create new parameters using the parameters 
                  ! specified in model command above. 
            new(ab*.2 ga_n*.0 ga_f*.5 ga_b ga_a*.0  
                pi_n*.4 pi_f*.4 pi_a*.2); ! true values assigned to the new parameters. 
            ab= a*b; ! SEM parameter ab. 
     

  ! from here, SEM parameters are translated into PS parameters. 
            ga_n = c; ! Eq. (25). 
            ga_f = b + c; ! Eq. (27) and constant effect. 
            ga_b =-b + c; ! Eq. (31). 
            ga_a = c; ! Eq. (26) and constant effect. 
            pi_f = a; ! Eq. (28) and monotonicity. 
            pi_a = alpha_m; ! Eq. (29) and monotonicity. 
            pi_n=1-pi_f-pi_a; ! Eq. (30) and monotonicity. 



Table 5 
5.c. Mplus input for PS_AIR model 
 
title:  Monte Carlo simulation using externally generated data.   
data:   file is alllist.dat;  type=montecarlo; 
variable:   names are  y z c3; 
              usev are y z m; 
define: if (c3==1 and z==1)  then m=0; 
        if (c3==1 and z==0)  then m=0; 
        if (c3==2 and z==1)  then m=1; 
        if (c3==2 and z==0)  then m=0; 
        if (c3==3 and z==1)  then m=1; 
        if (c3==3 and z==0)  then m=1; 
 
model: ! PS_AIR analysis model 
      y on z*0.2 (gamma_t); ! the overall (total) treatment assignment effect. 
      m on z*0.4 (pi_f); ! PS parameter pi_f. 
     [m*0.2] (pi_a); ! PS parameter pi_a. 
 
     model constraint: 
            new(gamma_f*0.5 pi_n*0.4 a*0.4 b*0.5 ab*0.2); 
            pi_n=1-pi_f-pi_a; ! Eq. (8). 
            gamma_f=gamma_t/pi_f; ! Eq. (10). 

   
! from here, PS parameters are translated into SEM parameters. 

            a=pi_f; ! Eq. (23) and monotonicity. 
            b=gamma_f; ! Eq. (22) and exclusion restriction. 
            ab=a*b; 
 



Table 9: When none of the SEM_BK and PS_AIR assumptions hold. 
 
9.a. Mplus input for data generation 
   
title: Data generation (true) model for Table 8 
       None of the assumptions of the SEM_BK and PS_AIR models hold.  
       See Table 2 for PS parameters considered in data generation. 
 
montecarlo: 
      names = y z; ! y is a continuous outcome. 
      nobservations = 500; ! N=500 in each data set. 
      nreps = 500; !500 data sets generated. 
      cutpoints = z(0); ! z is the intervention assignment. 
                        ! split at zero in the normal distribution. 
                        ! 50% control, 50% intervention. 
      genclasses = c(4); ! 4 principal strata are generated. 
      classes = c(1); ! analysis is done assuming one class. 
      repsave = all; ! all 500 data sets will be saved. 
      save = none*.dat; ! each data set has a name starting with none. 
 
model population: ! data generation model. 
      %overall% 
       [y*1];  ! generate the outcome. 
        y*1;   ! normally distributed with mean=1 and variance=1. 
       [z*1];  ! generate an intervention assignment variable. 
        z*0;   ! normally distributed with mean=0 and variance=1. 
               ! this variable is dichotomized (see cutpoints command).  
      y on z*0.15; ! the overall treatment assignment effect = 0.15. 
      
     [c#1*0.405465]; ! multinomial logit intercepts 
     [c#2*-0.69315]; ! to match true principal stratum proportions (0.3,0.4,0.1,0.2). 
     [c#3*0.693150]; 
 
      %c#1% ! never-improver 
          [y*1] ; ! mean outcome for never-improvers under the control condition. 
           y on z*-.1; ! treatment assignment effect for never-improvers = -0.1. 
      %c#2% ! backward-improver 
          [y*1] ; ! mean outcome for backward-improvers under the control condition. 
           y on z*-.4; ! treatment assignment effect for never-improvers = -0.4. 
      %c#3% ! forward-improver 
          [y*1.2] ; ! mean outcome for forward-improvers under the control condition. 
           y on z*.5; ! treatment assignment effect for forward-improvers = 0.5. 
      %c#4% ! always-improver 
          [y*1.6] ; ! mean outcome for always-improvers under the control condition. 
           y on z*.2; ! treatment assignment effect for always-improvers = 0.2. 
 
 Analysis: ! data generation model. 
      type = mixture ; ! data are analyzed assuming a mixture distribution. 
                       ! for Table 8, analyses done here are not necessary. 
                       ! necessary only to complete data generation in Mplus.  
 model: 
        %overall% 
       y on z*0.15 ; ! total effect of treatment assignment estimated. 
                  ! existence of 4 principal strata is ignored (see classes=1 command). 
 
 
 



Table 9 
9.b. Mplus input for SEM_BK model 
 
title: Monte Carlo simulation using externally generated data.   
data:   file is nonelist.dat; type=montecarlo; ! read in 500 data sets from 4.a 
variable: names are y z c4; ! c4 is principal stratum class, z is treatment assignment. 
          usev  are y z m;  ! observed mediator created based on c4 and z (see below). 
  define: if (c4==1 and z==1)  then m=0; 
          if (c4==1 and z==0)  then m=0; 
          if (c4==2 and z==1)  then m=0; 
          if (c4==2 and z==0)  then m=1; 
          if (c4==3 and z==1)  then m=1; 
          if (c4==3 and z==0)  then m=0; 
          if (c4==4 and z==1)  then m=1; 
          if (c4==4 and z==0)  then m=1; 
 
model: ! SEM_BK analysis model 
      y on z*-.1 (c); ! SEM parameter c. true values specified on the right side of *       
                    ! so that coverage rates can be calculated. 
      y on m*.3 (b); ! SEM parameter b. 
      m on z*.3 (a); ! SEM parameter a. 
     [m*0.2] (alpha_m); ! SEM parameter alpha_m. 
 
model constraint: ! this command lets us create new parameters using the parameters 
                  ! specified in model command above. 
            new(ab*.09 ga_n*-.1 ga_f*.5 ga_b*-.4 ga_a*.2  
                pi_n*.3 pi_f*.4 pi_a*.2); ! true values assigned to the new parameters. 
            ab= a*b; ! SEM parameter ab. 
     

  ! from here, SEM parameters are translated into PS parameters. 
            ga_n = c; ! Eq. (25). 
            ga_f = b + c; ! Eq. (27) and constant effect. 
            ga_b =-b + c; ! Eq. (31). 
            ga_a = c; ! Eq. (26) and constant effect. 
            pi_f = a; ! Eq. (28) and monotonicity. 
            pi_a = alpha_m; ! Eq. (29) and monotonicity. 
            pi_n=1-pi_f-pi_a; ! Eq. (30) and monotonicity. 



Table 9 
9.c. Mplus input for PS_AIR model 
 
title:  Monte Carlo simulation using externally generated data.   
data:   file is nonelist.dat;  type=montecarlo; 
variable:   names are  y z c4; 
            usev are y z m; 
define: if (c4==1 and z==1)  then m=0; 
          if (c4==1 and z==0)  then m=0; 
          if (c4==2 and z==1)  then m=0; 
          if (c4==2 and z==0)  then m=1; 
          if (c4==3 and z==1)  then m=1; 
          if (c4==3 and z==0)  then m=0; 
          if (c4==4 and z==1)  then m=1; 
          if (c4==4 and z==0)  then m=1; 
 
model: ! PS_AIR analysis model 
      y on z*0.15 (gamma_t); ! the overall (total) treatment assignment effect. 
      m on z*0.4 (pi_f); ! PS parameter pi_f. 
     [m*0.2] (pi_a); ! PS parameter pi_a. 
 
     model constraint: 
            new(gamma_f*0.5 pi_n*0.3 a*0.3 b*0.3 ab*0.09); 
            pi_n=1-pi_f-pi_a; ! Eq. (8). 
            gamma_f=gamma_t/pi_f; ! Eq. (10). 

   
! from here, PS parameters are translated into SEM parameters. 

            a=pi_f; ! Eq. (23) and monotonicity. 
            b=gamma_f; ! Eq. (22) and exclusion restriction. 
            ab=a*b; 



Table 9 
9.d. Mplus input for SEM_Mac (PS_Ig) model 
 
title:  Monte Carlo simulation using externally generated data.   
data:   file is nonelist.dat;  type=montecarlo; 
variable:   names are  y z c4; 
            usev are y z m mz; ! interaction parameter is included in the analyses. 
define: if (c4==1 and z==1)  then m=0; 
          if (c4==1 and z==0)  then m=0; 
          if (c4==2 and z==1)  then m=0; 
          if (c4==2 and z==0)  then m=1; 
          if (c4==3 and z==1)  then m=1; 
          if (c4==3 and z==0)  then m=0; 
          if (c4==4 and z==1)  then m=1; 
          if (c4==4 and z==0)  then m=1; 
          mz = m*z; ! interaction parameter is created based on m and z 
 
model: ! SEM_Mac analysis model 
      y on z*-.1 (c); ! SEM parameter c. true values specified on the right side of *       
                    ! so that coverage rates can be calculated. 
      y on m*.3 (b); ! SEM parameter b. 
      y on mz*.09 (d); ! SEM parameter d. 
      m on z*.3 (a); ! SEM parameter a. 
     [m*0.2] (alpha_m); ! SEM parameter alpha_m. 
 
model constraint:  
            new(ga_n*-.1 ga_f*.5 ga_b*-.4 ga_a*.2  
                pi_n*.3 pi_f*.4 pi_a*.2); ! true values assigned to the new parameters. 
            ab= a*b; ! SEM parameter ab. 
     

  ! from here, SEM parameters are translated into PS parameters. 
            ga_n = c; ! Eq. (25). 
            ga_f = b + c + d; ! Eq. (27). 
            ga_b =-b + c; ! Eq. (31). 
            ga_a = c + d; ! Eq. (26). 
            pi_f = a; ! Eq. (28) and monotonicity. 
            pi_a = alpha_m; ! Eq. (29) and monotonicity. 
            pi_n=1-pi_f-pi_a; ! Eq. (30) and monotonicity. 

 


